

Systèmes de transitions - Modélisation TLA+

Durée 1h45 - Documents autorisés

7 avril 2015

1 Questions de cours (2 points)

Soit D un ensemble. Soient $S \subseteq D$ et $f \in [D \rightarrow \text{Nat}]$ deux variables. Répondre aux questions suivantes en respectant la syntaxe TLA.

1. Donner un prédicat testant si au plus un entier x de S est tel que $f[x] = 0$.

$$\text{Cardinality}(\{x \in S : f[x] = 0\}) \leq 1$$

2. Donner une expression représentant l'ensemble des entiers x de S tels que $f[x]$ est l'élément maximum de l'ensemble $f(D)$.

$$\{x \in S : (\forall y \in D : f[x] \geq f[y])\}$$

$$\{x \in S : f[x] = \max(\{f[y] : y \in D\})\}$$

3. Donner une action ajoutant à S les entiers x de D tels que $f[x] = 0$.

$$S' = S \cup \{x \in D : f[x] = 0\}$$

$$S' = S \cup f^{-1}[0]$$

4. Donner une propriété temporelle exprimant que l'ensemble $f(S)$ n'est jamais vide.

$$\square(\{f[x] : x \in S\} \neq \emptyset) \text{ ou } \square(S \neq \emptyset)$$

2 Exercice (6 points)

Soit le module TLA fourni `Test.tla` définissant le système de transitions `Spec`.

1. Donner le graphe d'exécution correspondant
2. Les propriétés suivantes, exprimées en logique LTL ou CTL, sont-elles vérifiées (donner une justification informelle)?

(a) $\diamond \neg s_0$

(f) $\exists \square s_0$

(b) $\diamond s_1 \Rightarrow \diamond s_2$

(g) $\exists \diamond \exists \square s_1$

(c) $\square(s_1 \Rightarrow \diamond s_2)$

(h) $\forall \diamond s_3 \Rightarrow \forall \diamond s_1$

(d) $\square(s_0 \vee s_1 \vee s_3)$

(i) $\exists \diamond \forall \square(s_1 \vee s_2)$

(e) $\diamond \square \neg s_2 \Rightarrow \square(s_0 \vee s_3)$

(j) $\forall \square \exists \diamond s_1$

---- MODULE Test ----

EXTENDS Naturals

VARIABLES etat

TypeInvariant == etat \in { "s0", "s1", "s2", "s3" }


Trans0 == /\ etat = "s0" /\ etat' \in { "s1", "s3" }

Trans1 == /\ etat = "s1" /\ etat' = "s2"

Trans23 == /\ etat \in { "s2", "s3" } /\ etat' = "s1"

Spec == /\ etat = "s0" /\ [] [Trans0 \/ Trans1 \/ Trans23]_<<etat>>
 /\ WF_<<etat>>(Trans1)

=====



- | | |
|--|---|
| 2. (a) <i>non (bégaiement initial)</i> | (f) <i>oui (bégaiement initial)</i> |
| (b) <i>oui WF</i> | (g) <i>non WF</i> |
| (c) <i>oui WF</i> | (h) <i>oui : $\forall \diamond s_3$ est faux ($s_0 \rightarrow s_1^\omega$)</i> |
| (d) <i>non $\dots s_2^\omega$</i> | (i) <i>oui $\dots (s_1 \rightarrow s_2)^\omega$</i> |
| (e) <i>oui : $\square \diamond s_2 \vee \square (s_0 \vee s_3)$</i> | (j) <i>oui ($s_1$ toujours accessible)</i> |

3 Problème : détection de la terminaison (12 points ¹)

On considère un calcul concurrent constitué d'un ensemble de processus. Tant qu'un processus est actif, il peut réveiller des processus passifs. Par contre, un processus passif (ou en attente), ne peut bien sûr pas réveiller d'autres processus. La question est de détecter quand le calcul est fini, c'est-à-dire que tous les processus sont passifs. Comme il n'est pas possible d'évaluer globalement et instantanément la propriété que tous les processus sont passifs, on regarde un à un chaque processus. Pour cela, on utilise un parcours cyclique (analogue à un jeton circulant) qui passe successivement sur tous les processus. Le jeton ne progresse que quand un processus est passif.

Un tour où le jeton ne voit que des processus passifs ne suffit pas : on peut observer à un instant donné qu'un processus est passif, mais il peut être réveillé plus tard par un autre processus, qui sera lui aussi vu comme passif quand on le regardera. Considérons deux processus p_i et p_j ; lors de la visite de p_i , p_i est passif et p_j actif ; avant que le jeton ne parvienne à p_j , p_j réveille p_i et devient passif ; quand le jeton visite p_j , il le trouve passif ; mais il est faux de conclure que p_i et p_j sont tous deux simultanément passifs.

Pour éviter cela, le jeton colore en blanc les processus rencontrés passifs, et un processus qui devient actif se colore en noir. La détection de la terminaison a lieu quand le jeton fait un tour complet en ne rencontrant que des processus blancs.

Note : l'état actif/passif ne concerne que le calcul dont on veut détecter la terminaison. Un processus passif peut toujours agir vis-à-vis de la détection de la terminaison (par exemple, transmettre le jeton).

Un squelette du module est fourni en 3.6

3.1 Transitions

Définir les prédicats de transitions suivants, en respectant les règles énoncées ci-dessus :

1. Compléter `Init` pour l'initialisation de la variable `passif`, tel qu'elle puisse contenir n'importe quelle configuration de processus actifs/passifs.
 $passif \in [Processus \rightarrow BOOLEAN]$
2. Compléter `ReveillerAutreProc(i)`, tel que le processus actif i réveille un autre processus.

1. Toutes les questions valent autant, sauf la question 15 qui vaut double.

$$\begin{aligned}
\text{ReveillerAutreProc}(i) &\triangleq \\
&\wedge \neg \text{passif}[i] \\
&\wedge \exists j \in \text{Processus} \setminus \{i\} : \\
&\quad \wedge \text{passif}' = [\text{passif EXCEPT } ![j] = \text{FALSE}] \\
&\quad \wedge \text{couleur}' = [\text{couleur EXCEPT } ![j] = \text{Noir}] \\
&\wedge \text{UNCHANGED}\langle \text{fini}, \text{jeton} \rangle
\end{aligned}$$

3. Compléter `DetecterTerminaison(i)`, tel que le processus i détecte la terminaison du calcul. Le critère de détection est que le processus est blanc, qu'il a le jeton, que le jeton a visité N processus.

$$\begin{aligned}
\text{DetecterTerminaison}(i) &\triangleq \\
&\wedge \text{jeton}[i] = N \\
&\wedge \text{couleur}[i] = \text{Blanc} \\
&\wedge \text{fini}' = \text{TRUE} \\
&\wedge \text{UNCHANGED}\langle \text{couleur}, \text{jeton}, \text{passif} \rangle
\end{aligned}$$

4. `Next` : toutes les transitions possibles du problème modélisé.

$$\begin{aligned}
\text{Next} &\triangleq \\
&\exists i \in \text{Processus} : \\
&\quad \vee \text{ReveillerAutreProc}(i) \\
&\quad \vee \text{DetecterTerminaison}(i) \\
&\quad \vee \text{EnvoyerJeton}(i) \\
&\quad \vee \text{DevenirPassif}(i)
\end{aligned}$$

3.2 Équité

5. Pour les actions suivantes, expliquer informellement s'il est nécessaire de mettre de l'équité ou si cela est superflu :

- (a) `ReveillerAutreProc(i)`
- (b) `DevenirPassif(i)`
- (c) `DetecterTerminaison(i)`
- (d) `EnvoyerJeton(i)`

- (a) *ReveillerAutreProc : non (action applicative)*
- (b) *DevenirPassif : oui (sinon le calcul peut ne jamais se terminer) ou non (action applicative)*
- (c) *DetecterTerminaison : oui (sinon on pourrait ne jamais détecter la terminaison)*
- (d) *EnvoyerJeton : oui (sinon le jeton ne serait pas obligé de tourner)*

6. Donner l'équité minimale nécessaire pour qu'un processus ne reste pas toujours actif.

$$\forall i \in \text{Processus} : \text{WF}_{\text{vars}}(\text{DevenirPassif}(i))$$

7. Donner l'équité minimale nécessaire pour que le jeton tourne et visite tous les processus.

- (a) $\forall i \in \text{Processus} : \text{WF}_{\text{vars}}(\text{EnvoyerJeton}(i))$
- (b) $\forall i \in \text{Processus} : \text{WF}_{\text{vars}}(\text{EnvoyerJeton}(i)) \wedge \text{WF}_{\text{vars}}(\text{DevenirPassif}(i))$
- (c) $\forall i \in \text{Processus} : \text{SF}_{\text{vars}}(\text{EnvoyerJeton}(i)) \wedge \text{WF}_{\text{vars}}(\text{DevenirPassif}(i))$

Seule la 7c est bonne : dans la 7a, un processus ayant le jeton peut rester définitivement actif; dans la 7b, deux processus font ping-pong passif/actif sans que le jeton ne bouge jamais.

3.3 Spécification

Définir les propriétés suivantes (qui ne sont pas nécessairement vérifiées par le modèle) :

8. **TerminaisonCorrecte** : si la terminaison est détectée, alors tous les processus sont passifs.
 $\Box(\text{fini} \Rightarrow \forall i \in \text{Processus} : \text{passif}[i])$
9. **TerminaisonDétectée** : Si les processus sont tous passifs, alors la terminaison sera finalement détectée.
 $(\forall i \in \text{Processus} : \text{passif}[i]) \rightsquigarrow \text{fini}$
10. **DétectionStable** : si la terminaison est détectée, alors elle reste définitivement vraie.
 $\Box(\text{fini} \Rightarrow \Box \text{fini})$
11. **CouleurCorrecte** : un processus blanc est nécessairement passif.
 $\forall i \in \text{Processus} : \Box((\text{couleur}[i] = \text{Blanc}) \Rightarrow \text{passif}[i])$
12. **PassifDevientBlanc** : tout processus définitivement passif finit par devenir blanc.
 $\forall i \in \text{Processus} : (\Diamond \Box \text{passif}[i]) \Rightarrow \Diamond(\text{couleur}[i] = \text{Blanc})$
N'est vrai que si $WF(\text{DevenirPassif}) \wedge SF(\text{EnvoyerJeton})$
13. **TransmissionJeton** : le jeton n'est jamais transmis par un processus actif.
 $\forall i \in \text{Processus} : \Box((\text{jeton}[i]' \neq \text{jeton}[i] \wedge \text{jeton}[i]' = 0) \Rightarrow \text{passif}[i])$
 $\forall i \in \text{Processus} : \Box(\neg \text{passif}[i] \wedge \text{jeton}[i] \neq 0) \Rightarrow \text{UNCHANGED } \text{jeton}[i]$
13. **NoirRéinitialise** : un processus noir finit par remettre le jeton à 1.
 $\forall i \in \text{Processus} : \text{couleur}[i] = \text{Noir} \rightsquigarrow \text{jeton}[\text{suivant}(i)] = 1$ (* vicieux *)
N'est vrai que si $WF(\text{DevenirPassif}) \wedge SF(\text{EnvoyerJeton})$
13. **ProcDevientPassif** : tout processus actif finit par être passif.
 $\forall i \in \text{Processus} : \Box \Diamond \text{passif}[i]$
 $\forall i \in \text{Processus} : (\neg \text{passif}[i]) \rightsquigarrow \text{passif}[i]$
13. **CalculTermine** : le calcul se termine.
Selon le sens qu'on donne à la phrase :
 $\Diamond(\forall i \in \text{Processus} : \text{passif}[i])$
 $\Diamond \text{fini}$ ou $\Diamond \Box \text{fini}$ (avec **DétectionStable**)

3.4 Analyse du problème – preuve

14. En s'appuyant sur le prédicat d'état initial et les quatre actions, démontrer l'invariant **CouleurCorrecte**.

Soit $\text{CouleurCorrecte}(i) \triangleq ((\text{couleur}[i] = \text{Blanc}) \Rightarrow \text{passif}[i])$. Alors **CouleurCorrecte** $\triangleq \forall i : \Box \text{CouleurCorrecte}(i)$. On note $\text{next}(i)$ la conjonction des actions.


- (a) $\text{Init} \Rightarrow \forall i : \text{CouleurCorrecte}(i)$: car $\text{couleur}[i] = \text{Noir}$
- (b) $\forall i, j : i \neq j \wedge \text{CouleurCorrecte}(i) \wedge \text{next}(j) \Rightarrow \text{CouleurCorrecte}(i)'$: car *unchanged vars*[i]
- (c) $\forall i : \text{CouleurCorrecte}(i) \wedge \text{ReveillerAutreProc}(i) \Rightarrow \text{CouleurCorrecte}(i)'$: car le processus devient noir
- (d) $\forall i : \text{CouleurCorrecte}(i) \wedge \text{DevenirPassif}(i) \Rightarrow \text{CouleurCorrecte}(i)'$: car $\text{passif}[i]'$ est vrai
- (e) $\forall i : \text{CouleurCorrecte}(i) \wedge \text{DetecterTerminaison}(i) \Rightarrow \text{CouleurCorrecte}(i)'$: car *unchanged* $\text{passif}[i]$ et $\text{couleur}[i]$
- (f) $\forall i : \text{CouleurCorrecte}(i) \wedge \text{EnvoyerJeton}(i) \Rightarrow \text{CouleurCorrecte}(i)'$: car *unchanged* passif et $\text{passif}[i]$ (donc $\text{passif}[i]'$) et $\text{couleur}[i]' = \text{Blanc}$

3.5 Analyse du problème – exploration de l'espace d'états


Voici une portion du graphe d'exécutions (avec à la fois des états inaccessibles et des états manquants). Un état $PA/BN/01$ correspond à $\text{passif}[0] \wedge \neg \text{passif}[1] \wedge (\text{couleur}[0] = \text{Blanc}) \wedge (\text{couleur}[1] = \text{Noir}) \wedge (\text{jeton}[0] = 0) \wedge (\text{jeton}[1] = 1)$, c'est-à-dire :

- processus 0 : Passif, Blanc, sans jeton
- processus 1 : Actif, Noir, ayant le jeton, avec compteur = 1

On notera de manière distinctive les états où le booléen *fini* est vrai.



15. Construire le graphe complet (18 états accessibles).



16. Utiliser le graphe, en expliquant comment, pour vérifier la propriété **CouleurCorrecte**.

C'est un invariant \Rightarrow à vérifier pour tous les états accessibles du graphe.

Il n'y a aucun état où un processus est Actif et Blanc \Rightarrow Ok.

17. Utiliser le graphe, en expliquant comment, pour vérifier la propriété **TerminaisonDétectée**.

Depuis n'importe quel état PP, on atteint nécessairement un état où fini est vrai (hors bégaiement supprimé par l'équité faible sur EnvoyerJeton et DetecterTerminaison).

3.6 Module fourni : terminaison_misra.tla

MODULE *terminaison_misra*

EXTENDS *Naturals, FiniteSets*

CONSTANT *N* le nombre de processus

ASSUME $N > 1$ au moins 2 processus

Processus $\triangleq 0 \dots N - 1$

Blanc \triangleq "blanc"

Noir \triangleq "noir"

Couleur $\triangleq \{Blanc, Noir\}$

VARIABLES

couleur, couleur de chaque processus

passif, actif/passif pour chaque processus

jeton, contient le nombre de processus blancs visités en séquence, ou 0 si le processus n'a pas le *jeton*

fini indique si la terminaison est détectée

vars $\triangleq \langle couleur, passif, jeton, fini \rangle$

TypeInvariant \triangleq

$\square(\wedge couleur \in [Processus \rightarrow Couleur]$
 $\wedge passif \in [Processus \rightarrow \text{BOOLEAN}]$
 $\wedge jeton \in [Processus \rightarrow 0 \dots N]$
 $\wedge fini \in \text{BOOLEAN}$
 $)$

possedeJeton(i) $\triangleq (jeton[i] \neq 0)$ Le processus *i* a-t-il le *jeton* ?

suivant(i) $\triangleq (i + 1) \% N$ Site *suivant* pour le *jeton*

Init \triangleq

$\wedge couleur = [i \in Processus \mapsto Noir]$

$\wedge jeton = [i \in Processus \mapsto \text{IF } (i = 0) \text{ THEN } 1 \text{ ELSE } 0]$ *jeton* sur le site 0

$\wedge fini = \text{FALSE}$

\wedge À COMPLÉTER : initialisation de *passif*

ReveillerAutreProc(i) \triangleq le processus actif *i* réveille un autre processus

À COMPLÉTER

DevenirPassif(i) \triangleq le processus actif *i* devient *passif*

$\wedge \neg passif[i]$

$\wedge passif' = [passif \text{ EXCEPT } ![i] = \text{TRUE}]$

$\wedge \text{UNCHANGED } \langle couleur, jeton, fini \rangle$

DetecterTerminaison(i) \triangleq le processus *i* détecte la terminaison

À COMPLÉTER

EnvoyerJeton(i) \triangleq le processus *i* transmet le *jeton* au *suivant*

$\wedge possedeJeton(i)$ il a le *jeton*

$\wedge \neg(jeton[i] = N \wedge couleur[i] = Blanc)$ et la terminaison n'est pas encore détectée

$\wedge passif[i]$ et il est *passif*

$\wedge couleur' = [couleur \text{ EXCEPT } ![i] = Blanc]$

$\wedge jeton' = [jeton \text{ EXCEPT } ![i] = 0, ![suivant(i)] = (\text{IF } (couleur[i] = Blanc) \text{ THEN } jeton[i] + 1 \text{ ELSE } 1)]$

$\wedge \text{UNCHANGED } \langle passif, fini \rangle$

Next \triangleq À COMPLÉTER

Fairness \triangleq À COMPLÉTER

Spec \triangleq

$\wedge Init$

$\wedge \square[Next]_{\langle vars \rangle}$

$\wedge Fairness$